
Liquid Haskell上での並列プログラムの時間計算量解析

大喜 優，西崎真也

本研究では，純粋関数型言語 Haskellでの並列プログラミングに着目して，静的な時間計算量の解析を試みる．Liquid Haskell

は，Haskell に篩型を追加して，SMT ソルバによる検証を行うプログラム検証器である．本研究では，Liquid Haskell を用
いた既存の逐次時間計算量解析手法を，並列プログラムへと拡張する．具体的には，時間計算量を解析する 2 つの関数を実
装した．1つは，2要素のタプルをプロセッサ数に応じて逐次または並列に実行する関数である．もう 1つは，リストの各要
素を並列に実行する関数である．提案した実装を，いくつかの例に適用することで，時間計算量の解析が適切に行われている
ことを確認した．また，実際に並列プログラムを実行して経過時間の計測を行い，解析した時間計算量との比較を行った．

1 はじめに
1. 1 並列プログラムと時間計算量
近年，シングルコアプロセッサの性能が頭打ちを迎え

たことで，マルチコアプロセッサを用いた並列計算によ
る高速化が重要性を増してきている．しかし，並列プロ
グラムを意図した性能となるよう記述することは，容易
ではない．適切な並列化を行えないと，逐次プログラム
と実行時間がさほど変わらなかったり，むしろより遅く
なってしまう場合もある．
実行時間の測定は，並列プログラムの場合においても，

効率良く動作していることの確認に有効な手段である．し
かし，並列化の実行時間は，入力に加えて実行する環境
（プロセッサの個数など）によっても変わってくるため，
あらゆる場合に対して想定通りに動くことを保証するの
は困難である．また，実行時間の測定は，ソースコード
の一部箇所を変えただけで測り直しとなってしまう場合
もある．時間計算量は，プログラムの性能を見積もる上
で重要な指標であり，並列プログラムにおいても同様で
ある．プロセッサ数を変数とする時間計算量により，プ
ロセッサ数を増加させることでどのように実際のステッ
プ数を減らすことができるかを判断することができる．

1. 2 Haskellと並列プログラミング
Haskell [2, 9, 12] は，純粋関数型のプログラミング言
語である．以下 Haskell に関して述べる際は，最も一般
的に用いられているコンパイラである Glasgow Haskell

Compiler（以下 GHC）[1]を用いているとする．
Haskell [10, 11, 13, 16] は，並列プログラミングに対応
した言語でもある．Haskellで記述した並列プログラムか
ら得られる結果は決定的であり，競合条件やデッドロック
を引き起こすことはない．また，同期や通信という概念
は暗黙のうちに扱われる．プログラマは並列化を期待す
る場所をソースコード中に示し，実行時システムによっ
て並列化するかの判断が行われる．
1. 2. 1 型クラス
型クラスは，多相型において，ある特定の計算を行え
る型を指定するための概念である．ある型クラスの指定
した計算，すなわち関数を実装した型は，その型クラス
のインスタンスとなることができる．多相型を引数に取
る関数において，型変数が型クラスのインスタンスであ
ることを要求することで，受け取る型が特定の計算を行
えることを強制できる．
例えば，型クラス Num のインスタンスとなる型は，

(+)や(-)，(*)などといった数値なら行えるべき演算を
扱える型である．具体的には，Int，Float，Integerなど
が該当する．
また，図 1関数 sumにおいて，型クラス Numを満た
す型 aのリスト [a]を引数として指定すると，リストの値
を全て(+)で足し合わせた型 aの値を返す関数を実装する
ことができる．



1 sum :: Num a => [a] -> a

2 sum [] = 0

3 sum (x:xs) = x + sum xs

図 1 関数 sum

1 return :: a -> m a

2

3 (>>=) :: forall a b. m a -> (a -> m b) -> m b

図 2 関数 return と(>>=)の型

1 data Maybe a = Nothing | Just a

図 3 Maybe モナド

1. 2. 2 モナド
モナド [15]とは，計算の状態や入出力等を表現するた

めの汎用的な構造である．あるモナドは，計算の対象で
あるデータ型と，計算値の生成関数，計算を合成する関
数から構成される．
Haskell では，モナドは return 関数（生成）と(>>=)

関数（合成）を持つ型クラスとして実装される．対象の
データ型にこれらの関数を定義し，モナドのインスタン
スとすることで，モナドとしてデータ型に計算を行うこ
とができる．図 2に示すのが，それぞれの型である．
return関数は，値からモナドの値を生成する関数であ

る．(>>=)関数は，モナドに包んだ型 a の値と，型 a の
値を取ってモナドに包んだ型 bの値を返すような関数を
受け取り，それらを合成してモナドに包んだ型 bの値を
返す関数である．
モナドの例としては，Maybeモナドや IOモナドが挙

げられる．
Maybeモナドは，値に，存在しない場合を加えて計算

結果として保持するモナドである．Maybeモナドを返す
ことで，計算結果の値か，その計算が失敗したという情
報のいずれかを返す関数を表現できる．(>>=)関数を用い
て，失敗するかもしれない計算を連続で行い，全て成功
した場合のみ最終的に Justの値を持つ，などの処理を書
くことができる．
IOモナドは，副作用を保持するモナドである．Haskell

の関数は本来純粋であるが，IOモナドを用いることで，
入出力のような純粋でない処理を表現することができる．

1. 3 Liquid Haskellにおける時間計算量解析
1. 3. 1 篩型
篩型 (Refinement Types) [17] とは，特定の述語を満

たすような値を分類する型のことである．次の例は，0で

1 (declare-const x Int)

2 (declare-const y Int)

3 (assert (= (+ x y) 10))

4 (assert (= (+ x (* 2 y)) 20))

5 (check-sat)

6 (get-model)

図 4 SMT ソルバで連立方程式を解く入力

1 {-@ safeDiv :: Int -> { v:Int | v /= 0 } -> Int @-}

2 safeDiv x y = x `div` y

3

4 main = print $ safeDiv 3 0

図 5 安全な div 関数の例

ないという述語を満たす整数の篩型である．
{v ∈ Z | v ̸= 0}

1. 3. 2 SMTソルバ
SMT(Satisfiability Modulo Theories) [4,21]とは，あ
る理論に関する一階述語論理式の充足可能性を決定する
問題である．SMTソルバとは，SMT問題を解くような
ソルバのことを言う．SMTソルバは，命題論理の部分は
SATソルバ，整数については整数の理論に関するソルバ
が処理するように作られている．代表的な SMTソルバ
としては，z3 [5]が挙げられる．
例えば x+ y = 10と x+2y = 20の 2つの等式からな
る整数変数 x,yの連立方程式は，SMTソルバに図 4のよ
うな入力を与えることで解くことができる．
1. 3. 3 Liquid Haskell

Liquid Haskell [3]は，Haskellに篩型を追加し，SMT

ソルバによる静的な検証を可能にした拡張である．Liquid
Haskellの検査を通過したプログラムは，GHCでもコン
パイル可能であることを保証されている．
Liquid Haskellの検査では，まず篩型から部分型の関
係を取り出す．次に部分型の関係を論理式に変換し，そ
の論理式が真であるかを SMTソルバによって確かめる．
Liquid Haskellでは，Haskellのソースコード中にコメ
ントとして篩型を記述する．例えば，0除算を発生させな
い安全な除算関数 safeDivは，図 5のように定義し，利
用できる．
Liquid Haskellは，第二引数が 0でないという篩型の
制約から，このプログラムが安全でない（unsafeである）
ことを報告する．
Liquid Haskellでは，関数の引数の型に述語を加える
ことで，関数の事前条件を表現できる．また，関数の返
り値の型に述語を加えることで，関数の事後条件を表現



1 {-@ data Tick a = Tick { tcost :: Int, tval :: a } @-}

2 data Tick a = Tick { tcost :: Int, tval :: a }

図 6 Tick モナドのデータ型

1 {-@ reflect return @-}

2 {-@ return :: x:a

3 -> { t:Tick a | x == tval t && 0 == tcost t } @-}

4 return :: a -> Tick a

5 return x = Tick 0 x

6

7 {-@ reflect >>= @-}

8 {-@ (>>=) :: t1:Tick a -> f:(a -> Tick b)

9 -> { t:Tick b |

10 tval (f (tval t1)) == tval t &&

11 tcost t1 + tcost (f (tval t1)) == tcost t }

12 @-}

13 infixl 4 >>=

14 (>>=) :: Tick a -> (a -> Tick b) -> Tick b

15 Tick m x >>= f = let Tick n y = f x in Tick (m + n) y

図 7 Tick モナドの return 関数，(>>=) 関数

できる．さらに，データ構造の型に述語を加えることで，
そのデータ構造で常に成り立つ不変条件を表現できる．
このように，Liquid Haskellで検査されたプログラム

は，指定した事前条件や事後条件，不変条件や満たして
いることが保証される．一方で，SMTソルバの推論の限
界から，篩型の条件を満たしていても推論されずに安全
でないと判定される場合もある．
1. 3. 4 Liquid Haskellにおける時間計算量解析
論文 [6]で Handleyらは，Liquid Haskellを用いて時

間計算量を静的に解析する手法を提案した．この手法で
は，リソース使用量を記録するデータ型として，Tickモ
ナドを定義する．
Tickモナドは，計算結果の値とともに，動作ステップ

数も保持するするようなモナドである．Tickモナドでは，
tcostが動作ステップ数，tvalが計算結果の値を表す．操
作を行う場合に動作ステップ数を表す tcost の値も適切
に加算することで，操作に必要な動作ステップ数を記録
することができる．
Tick モナドを構成する return 関数と (>>=) 関数の定

義は，図 7のようになる．
return関数は，値を取って tcostが 0の Tickモナドを

返す．
(>>=)関数は，第一引数の tcostであるmと，第二引数

fに第一引数の値 xを適用した結果の tcostである nを加
算した結果の (m+n) を，返される Tick モナドの tcost

とする．
Liquid Haskell を用いた時間計算量解析には，2 つの

1 {-@ eqBind :: n:Int -> t1:Tick a

2 -> f:(a -> { tf:Tick b | n == tcost tf })

3 -> { t:Tick b |

4 tval (f (tval t1)) == tval t &&

5 tcost t1 + n == tcost t }

6 @-}

7 eqBind :: Int -> Tick a -> (a -> Tick b) -> Tick b

図 8 eqBind 関数

1 {-@ (++) :: xs:[a] -> ys:[a]

2 -> { t:Tick {

3 zs:[a] | length zs == length xs + length ys

4 } | tcost t == length xs }

5 @-}

図 9 (++) 関数

手法がある．Liquid Haskellを用いた自動証明と，Proof

Combinators を用いた外部での手動の定理証明である．
自動証明では，手動で証明をする必要がないという利点
がある．一方で，自動証明では型レベルの計算が必要な
ことから，(>>=)関数を用いただけでは推論できないとい
う問題がある．そこで，tcost (f (tval t1))を引数の整数
に置き換えた，eqBind関数（図 8）を用いて推論を行う．
tcostは動作ステップ数以外のリソース使用量，例えば

Thunkの個数などを記録することもできる．

1. 4 研究目的
本研究では，Handleyらが提唱した Liquid Haskellを
用いた時間計算量解析を，並列プログラムに拡張した実
装を提案する．提案した実装を用いて，実際にこれを適
用したいくつかの並列プログラムを紹介し，その時間計
算量を解析する．また，実際に並列プログラムを実行し
て経過時間の計測を行い，解析したステップ数との比較
を行う．

1. 5 関連研究
白水ら [20] は，定理証明システム Coq を用いた計算
量解析手法 [14]の並列プログラムへの拡張手法を提案し
た．図 10のように定義されるモナド Cは，値の型 Aと，
型 Aの値とプロセッサ数から成り立つ計算量の性質を表
す命題を返す述語 Pからなる．並列に拡張するにあたっ
て，並列度を表すプロセッサ数の 2の対数の引数，並列
呼び出しを指定する書式，並列度に応じた計算量算出を
追加している．
計算量に関する操作を型の内部に記述することで，Coq

から OCaml に計算量の性質を保ってコード抽出を行え
る．一方で，Coqでの証明は，手動で行う必要がある．



1 Definition C (A:Set) (P:A -> nat -> Prop) : Set :=

2 {a : A | exists (an:nat), (P a an)}.

図 10 モナド C の定義

Hoffmann ら [7] は，並列プログラムのコストを静的
に自動解析する手法を提案した．全体の計算時間の合計
である workと最も長い計算の時間である depthから並
列プログラムのコストを評価できることに着目している．
また，提案した解析手法を Resource Aware ML(RaML)

に実装することで，その実用性を確認している．

2 並列プログラムの時間計算量解析
本章では，まず並列プログラムの計算機モデルを考え

る．次に，本研究で実装したライブラリ関数を紹介する．
その後，時間計算量の解析対象となる具体的な並列処理
を，実装した関数を用いて実装する．

2. 1 PRAMモデル
時間計算量や空間計算量について考える場合は，前提と

なる計算機モデルを考える必要がある．逐次プログラムで
は，チューリング機械やランダムアクセス機械（random

access machine，以下 RAMと略す）が採用されること
が多い．
並列ランダムアクセス機械（parallel random access

machine, 以下 PRAMと略す）[18] [8] [19]は，並列計算
のモデルとして最もよく用いられているものの 1つであ
る．PRAMは，局所メモリを持つ RAMであるプロセッ
サの列と，全てのプロセッサがアクセスできる共有メモ
リから構成される．PRAMでは，各プロセッサは各命令
（ステップ）を 1単位時間で実行し，命令実行サイクルは
同期しているものとする．PRAMが停止したとは，全て
のプロセッサが停止したことを言う．
並列プログラムの時間計算量は，PRAMが停止するま

でに経過した時間で定義される．各ステップは 1単位時間
で実行されることから，時間計算量は全体を 1つのプロ
セッサとして見た場合のステップ数としても解釈できる．
一方で，並列プログラムの仕事量は，あるプログラム

で各プロセッサが実行する命令数の合計である．仕事量
と時間の関係を work-timeパラダイムと呼ぶ．
以降，ある入力 n を取る並列アルゴリズムに関して，

プロセッサ数を pとしたときステップ数が T (p, n)，仕事
量がW (n)として表記する．
一般にW (n) ≤ pT (p, n)が成り立つ．等号は，全ての

プロセッサが PRAM が停止するまで命令を実行した場
合である．p = 1 の場合，すなわち逐次プログラムとし

1 {-@ evalPair :: t12:(Tick a, Tick b) -> lbp:Nat

2 -> { t:Tick (a, b) |

3 tcost t == if lbp == 0

4 then (tcost (fst t12) + tcost (snd t12))

5 else tcost (fst t12) &&

6 tval t == (tval (fst t12), tval (snd t12))}

7 @-}

8 evalPair :: (NFData a, NFData b) =>

9 (Tick a, Tick b) -> Int -> Tick (a, b)

図 11 evalPair 関数

て実行した場合はW (n) = T (1, n)である．

2. 2 並列処理関数の紹介
本研究ではHandleyら [6]の手法を拡張したため，Tick
モナドの定義も同一である．tvalは型の値であり，本研
究では tcost が先ほど定義した並列プログラムの時間計
算量を示す．ステップ数と時間計算量は同じものを指す
が，加算処理を行う場合はステップ数として解釈した方
が自然なため，主に途中過程ではステップ数として扱う．
本研究では，大きく分けて 2つのライブラリ関数を設
計・実装した．ここでは簡単な紹介に留め，詳細に関し
ては実装の章で説明する．
2. 2. 1 evalPair関数
evalPair関数は，Tick型 2要素から構成されるタプル

(Tick a, Tick b) とプロセッサ数に関する情報を受け取
り，タプルを値とする Tick型である Tick (a, b) を返す
ような関数である．
この際，引数のタプルには存在する 2つの Tickは，そ
れぞれが tcostを持つ．これらは，Tickの値を単体で実
行（評価）した場合のステップ数を示している．
evalPair関数では，プロセッサ数の 2の対数の値を引
数で渡すことで，プロセッサ数が 1ならば逐次計算，1よ
り多いならば並列計算が行われる．
逐次計算では，図 12のように 1つ目の要素，2つ目の
要素と同じプロセッサで順に評価される．そのため，2つ
の要素のステップ数である tcostが加算される．
一方，並列計算では，図 13のように並列に評価を行う．
つまり，空いている別のプロセッサに 1つ目の処理を任
せ，2つ目の処理を元のプロセッサで行う．両方の処理が
終わってから，次の処理が行われる．並列に評価して同
期する場合，全体としてのステップ数は 1つ目と 2つ目
のうちステップ数が多い方を取る．（ただし，ここでは 1

つ目の要素の方が tcost が大きくなるように渡すことを
仮定している．詳細は実装の章にて）
また，補助関数として (>>=/) 関数（図 14）を定義し
ておく．(>>=/)関数は，次のように定義される関数であ
る．引数は(>>=)と同一だが，篩型における制約が異な



図 12 ペアの逐次計算

図 13 ペアの並列計算

1 {-@ (>>=/) :: t1:Tick a

2 -> f:(a -> { tf:Tick b | tcost tf == 1 })

3 -> { t:Tick b |

4 tval (f (tval t1)) == tval t &&

5 tcost t1 + 1 == tcost t }

6 @-}

7 infixl 4 >>=/

8 (>>=/) :: Tick a -> (a -> Tick b) -> Tick b

図 14 (>>=/) 関数

1 {-@ parList :: c:Nat

2 -> { xs:[{ t1:Tick a | tcost t1 <= c }] | xs /= [] }

3 -> { p:Nat | p >= length xs }

4 -> { t:Tick {

5 os:[a] | length xs == length os

6 } | tcost t == c }

7 @-}

8 parList :: (NFData a) =>

9 Int -> [Tick a] -> Int -> Tick [a]

図 15 parList 関数

る．(>>=/)関数では，引数 fの返り値の Tickの tcostが
1 であることを制約とする．これにより，tcost (f (tval

t1))の推論を 1で置き換えることができ，SMTソルバで
の推論が可能である．
2. 2. 2 parList関数
parList 関数は，Tick 型で構成されるリスト [Tick a]

図 16 リストの並列計算

1 {-@ chunk :: { n:Nat | n /= 0 } -> { m:Nat | m /= 0 }

2 -> { xs:[a] | length xs == n * m }

3 -> { yss:[ { ys:[a] | length ys == n } ] |

4 length yss == m }

5 / [m]

6 @-}

7 chunk :: Int -> Int -> [a] -> [[a]]

図 17 chunk 関数

1 {-@ mapParList :: { n:Nat | n > 0 }

2 -> { xs:[a] | length xs == n }

3 -> c:Nat

4 -> (a -> { t:Tick b | tcost t <= c })

5 -> { t:Tick { os:[b] | length xs == length os } |

6 tcost t == c }

7 @-}

8 mapParList :: (NFData b) =>

9 Int -> [a] -> Int -> (a -> Tick b) -> Tick [b]

図 18 mapParList 関数

とコスト，プロセッサ数の情報を受け取り，リストを値
とする Tick型である Tick [a] を返すような関数である．
parList関数では，図 16のように配列を構成する各要
素の処理を並列に評価する．
parList関数の引数として受け取る Tick型のリストは，
引数で受け取る cost以下の tcostであることを篩型の制
約としている．また，引数で受け取ったプロセッサ数は
配列の長さ以上であることを篩型の制約としている．よっ
て受け取ったプロセッサを使って配列の各要素を並列に
評価することができる．そのため，全体の tcost も引数
として受け取った costとすることができる．
parList 関数を使用する際の補助として，chunk 関数
や mapParList関数を実装した．それらの簡単な紹介も
行う．
chunk 関数は，リストの分割数を受け取り，受け取っ
たリストを分割してリストのリストとして返すような関
数である．
mapParList関数は，リストに対して渡した関数を並列
に実行する関数である．先ほどの parList 関数を内部で
呼び出し，並列化を行う．
以降は，これらの関数を用いて並列プログラムの時間
計算量解析を行う．



1 {-@ fib :: n:Nat -> { v:Nat | v >= n } @-}

2 fib :: Int -> Int

3 fib 0 = 1

4 fib 1 = 1

5 fib n = fib (n -1) + fib (n -2)

図 19 fib 関数

1 {-@ fibPar :: n:Nat -> lbp:Nat

2 -> { t:Tick Int | (tcost t == fibCost n lbp) } @-}

3 fibPar :: Int -> Int -> Tick Int

4 fibPar 0 _ = wait 1

5 fibPar 1 _ = wait 1

6 fibPar n 0 =

7 ((x, y) `evalPair` 0)

8 >>=/ (\(a, b) -> wait (a + b))

9 where

10 x = fibPar (n -1) 0

11 y = fibPar (n -2) 0

12 fibPar n lbp =

13 ((x, y) `evalPair` lbp)

14 >>=/ (\(a, b) -> wait (a + b))

15 where

16 x = fibPar (n -1) (lbp -1)

17 y = fibPar (n -2) (lbp -1)

図 20 fibPar 関数

2. 3 並列フィボナッチ関数
2. 3. 1 フィボナッチ関数
フィボナッチ数列とは，以下の漸化式で定義される数

列である．
F0 = 1

F1 = 1

Fn+2 = Fn+1 + Fn (n ≥ 0)

まず，入力 nを受け取り，Fn を返す関数 fibの実装を
以下に提示する．
入力 nが 0か 1の場合は，1を返す．
入力 n > 1の場合，fib(n−1)と fib(n−2)の和を返す．
2. 3. 2 並列フィボナッチ関数
fib関数（図 19）は，再帰的に fib関数を呼び出す部分

を並列に行うことで，容易に並列化が可能である．時間
計算量を tcostとして保持し，並列化を施した fibPar関
数について考える．
fibParは，入力 nと，プロセッサ数の 2の対数を表す

lbpを受け取り，Fn を値とするような Tick型を返す関数
である．fibPar関数の実装は図 33のようになっている．
入力が 0または 1の場合，値は 1である．wait関数は，

値を受け取り，tcostに 1を割り当てた Tickを返す関数
である．
入力 n > 1の場合，値としては fib(n− 1)と fib(n− 2)

1 {-@ fibCost :: n:Nat -> lbp:Nat -> Nat @-}

2 fibCost :: Int -> Int -> Int

3 fibCost 0 _ = 1

4 fibCost 1 _ = 1

5 fibCost n lbp

6 | n <= lbp + 1 = n

7 | otherwise = 2 * fib (n - lbp) -1 + lbp

図 21 fibCost 関数

を加算したい．そのため，xには fib(n − 1)を計算する
ように，y には fib(n− 2)を計算するように fibPar関数
を呼び出した結果を代入する．このとき，プロセッサ数
が 2以上（すなわち，プロセッサ数の 2の対数が 1以上）
の場合は，プロセッサ数の 2の対数から 1を引いた値を
fibPar関数の引数として渡す．この時点では，xと y は
まだ評価されていない．
その上で，タプル (x, y)とプロセッサ数の 2の対数を

evalPair 関数に渡す．evalPair 関数では，プロセッサに
空きがある，すなわちプロセッサ数が 2以上の場合（こ
の処理を実行中のプロセッサに加えて 1つある場合）は
並列に処理が行われ，プロセッサ数が 1の場合（実行中
のプロセッサ以外に空いていない場合）は逐次処理が行
われる．xと y の処理内容は fibPar関数の呼び出しなの
で，再帰的に処理が繰り返される．
逐次処理でも並列処理でも，evalPairの結果は eqBind

関数の第 2引数となる．eqBind関数によって，ペアの和
を取って wait関数に渡して返す関数に xと yの値のペア
が渡され，加算された結果の Tick型が返ってくる．wait

関数では tcostとして 1が指定されるので，全体の tcost

は，evalPair関数の結果の Tick型の tcostに，さらに 1

が加算された値となる．
ただし，Liquid Haskellの推論の仕様上，lbp = 0の場
合とそうでない場合とで明示的に場合分けが必要になっ
てしまっている．
2. 3. 3 並列フィボナッチ関数の時間計算量
fibCost は fibPar の時間計算量を表す関数であり，図

21のように定義される．
入力 nと lbpを用いて，fibCostが返す値を T (n, lbp)

として以下のように表せる．

T (0, lbp) = 1,

T (1, lbp) = 1,

T (n, lbp) = n, (n ≤ lbp + 1)

T (n, lbp) = 2 · fib(n− lbp)− 1 + lbp,

(otherwise).

したがって，十分大きな nにおいて lbp が lbp < nとな



る場合は，
T (n, lbp) = O

((
1 +

√
5

2

)n−lbp
)
.

となり，十分大きな nにおいて lbp が，n ≤ lbp +1とな
る場合は，

T (n, lbp) = O(n)

となる．
O記法は，f(n) = O(g(n))と表記したとき，ある正数

kと n0 に対して任意の n > n0 に対し，|f(n)| ≤ k · g(n)
が成り立つことを表す．すなわち，f は漸近的に gによっ
て上から抑えられることを意味する．
上の結果からは，並列数が少なければ指数時間がかか

るが，充分並列性が確保されたならば，線形時間で計算
が終わることがわかる．一方で，lbpが nより大きい場合
というのは，プロセッサ数の 2の対数が nより大きい場
合であり，およそ 2n のプロセッサ数が必要であるので，
現実的ではない．
それぞれの場合について，上の時間計算量が正しいこ

とを簡単に説明する．
n = 0または n = 1の場合
wait 関数の tcost が 1 であるため，lbp の値に関わら

ずステップ数は 1となる．
1 < nかつ n ≤ lbp+ 1の場合
n ≤ lbp+ 1の場合とは，プロセッサ数が充分多く，並

列化したいだけ並列化できる場合である．fibPar関数は
引数 n を受け取ると，fibPar 関数に n − 1，n − 2 を渡
して並列に実行する．このうち，よりステップ数が多い
のは n − 1を渡した場合である．ステップ数が多い方を
追うと，最終的に n = 1で停止する．nから 2までは加
算の際，1では値を返す際に wait関数により 1つずつス
テップ数が増えるので，tcostの値 nと一致する．
一方，1 < nかつ lbp ≤ n の場合に関しては，lbp = 0

の場合とそうでない場合とで場合分けして考える．
1 < nかつ lbp = 0の場合
この場合，プロセッサ数は 1である．n = 0と n = 1

の場合を除くと n > 1であり，また n ≤ lbp + 1を満た
さないので，4つ目の場合となる．lbp = 0を代入すると，
c = 2× fib(n)− 1となる．
この値になることの直感的な説明を行う．lbp = 0かつ

n > 1の場合，fib(n)の計算結果は，以下のように fib(0)

または fib(1)の和となる．

図 22 fib(5,2) の実行例

fib(2) = fib(1) + fib(0) = 2

fib(3) = fib(2) + fib(1)

= fib(1) + fib(0) + fib(1) = 3

fib(4) = fib(3) + fib(2)

= fib(2) + fib(1) + fib(1) + fib(0) + fib(1)

fib(n) =

fib(n)︷ ︸︸ ︷
fib(1) + . . .+ fib(1)

fib(0) = fib(1) = 1のため，fib(n)を構成する fib(0)と
fib(1)の和は fib(n)となる．fibParの計算では，fib(0)ま
たは fib(1)を返す際に tcostに 1を指定して返す．また，
fib(n− 1)と fib(n− 2)を加算する際にも tcostに 1を追
加する．これら以外で tcostが加算されることはない．し
たがって，値を返す際に wait 関数でステップ数 fib(n)，
加算の際にステップ数 (fib(n)− 1)が追加されるため，合
計のステップ数は 2× fib(n)− 1である．
1 < nかつ 0 < lbp ≤ nの場合
この場合は，プロセッサが余っていて並列化できると
ころまでは並列化して，各プロセッサにタスクが行き渡っ
たら逐次処理を行う．
例として，図 33は fibPar(5,2)を実行した場合の処理の
流れを表している．fibPar(5,2)は，22個のプロセッサ上で
fib(5)の計算を行う．fibPar(5,2)では並列に fibPar(4,1)

と fibPar(3,1)を呼び出す．fibPar(4,1)では fibPar(3,0)

と fibPar(2,0)を並列に呼び出し，これら fib(3)と fib(2)

の計算が各プロセッサで並列に行われる．fibPar(3,1)も
同様で，fib(2) と fib(1) の計算をそれぞれのプロセッサ
で並列に行う．ステップ数が最も多いのは fib(3) であ
り，上述したように 2× fib(3)− 1である．よって，fib-

Par(4,1) のステップ数は加算処理でステップ数に 1 を
加えた 2 × fib(3) となる．したがって fibPar(5,2) のス
テップ数は加算処理でさらにステップ数に 1 を加えた
2× fib(3) + 1 = 2× fib(5− 2)− 1 + 2となる．
以上が，並列 fib関数の時間計算量の解析と，その直感
的な説明である．



1 {-@ pow2 :: n:Nat -> { v:Nat | v >= n } @-}

2 pow2 :: Int -> Int

3 pow2 0 = 1

4 pow2 n = pow2 (n -1) + pow2 (n -1)

図 23 pow2 関数

1 {-@ pow2Par :: n:Nat -> lbp:Nat

2 -> { t:Tick Int | (tcost t == pow2Cost n lbp) } @-}

3 pow2Par :: Int -> Int -> Tick Int

図 24 pow2Par 関数

1 {-@ reflect pow2Cost @-}

2 {-@ pow2Cost :: Nat -> lbp:Nat -> {v:Nat | v >= 1} @-}

3 pow2Cost :: Int -> Int -> Int

4 pow2Cost 0 _ = 1

5 pow2Cost n lbp

6 | n <= lbp = n + 1

7 | otherwise = pow2 (n + 1 - lbp) -1 + lbp

図 25 pow2Cost 関数

2. 4 並列に 2の冪乗を計算する関数
pow2関数とは，与えられた nに対して 2n を計算する

関数のことを意味する．以下が，pow2関数の実装の一例
である．より効率のよい実装は存在するが，今回は並列
化の紹介のため図 23のような実装とする．
並列で pow2を計算する pow2Par関数も，fibPar関数

と同様に，再帰的に呼び出す処理を並列化することで図
24のように定義できる．
pow2Cost関数は pow2Par関数の時間計算量を表す関

数であり，図 25のように定義される．
pow2Cost が返す値は，入力 n と lbp を用いて以下の

式で表せる．

T (n, lbp) =



1 = O(1)

(n = 0)

n+ 1 = O(n)

(0 < n ∧ n <= lbp)

2n+1−lbp − 1 + lbp = O
(
2n−lbp

)
(otherwise)

(1)

これが pow2Par 関数の時間計算量として正しいこと
は，fibPar関数と同様に考えることでわかる．
この結果は，論文 [20]における並列 pow2関数の時間

計算量解析の結果とも一致する．ただし，pow2関数の計
算の際には加算する 2 つの値が同じ式なので，GHC の
最適化によってこの通りの時間計算量にはならないこと
に注意が必要である．

図 26 完全二分木の例

1 {-@ measure getHeight @-}

2 {-@ getHeight :: PBT a -> Nat @-}

3 getHeight :: PBT a -> Int

4 getHeight Leaf = 0

5 getHeight (Node _ h _ _) = h

6

7 {-@

8 data PBT a

9 = Leaf

10 | Node

11 { key :: a,

12 h :: Nat,

13 l :: { v:PBT a | getHeight v == h - 1 },

14 r :: { v:PBT a | getHeight v == h - 1 }

15 }

16 @-}

17 data PBT a

18 = Leaf

19 | Node

20 { key :: a,

21 h :: Int,

22 l :: PBT a,

23 r :: PBT a

24 }

25 deriving (Show)

図 27 完全二分探索木の定義

2. 5 完全二分木の並列走査
ある木の高さとは，その木の根から葉までの経路のう
ち，最も長いもののことを言う．完全二分木とは，全て
のノードの左右部分木の高さが等しいような二分木のこ
とである．
Liquid Haskellでは，図 27のように完全二分木を定義
できる．
PBT は，完全二分木を表すデータ型である．Haskell

上では単なる高さと keyを保持した二分木となるが，篩
型の制約によって完全二分木であることを保証している．
getHeight 関数は，与えられた二分木の高さを得る関数
である．左右の部分木の高さが両方とも二分木の高さ h

より 1小さい値であることを制約としているので，PBT

は完全二分木となる．ただし，この定義では葉は値を持
たないことに注意が必要である．



1 {-@ sumPBT :: { p:PBT Nat | getHeight p /= 0 }

2 -> lbp:Nat

3 -> { t:Tick Int |

4 (tcost t == pow2Cost (getHeight p - 1) lbp) }

5 @-}

6 sumPBT :: PBT Int -> Int -> Tick Int

7 sumPBT (Node key 1 _ _) _ = wait key

8 sumPBT (Node key _ l r) 0 =

9 ((x, y) `evalPair` 0)

10 >>=/ (\(a, b) -> wait (a + b + key))

11 where

12 x = sumPBT l 0

13 y = sumPBT r 0

14 sumPBT (Node key _ l r) lbp =

15 ((x, y) `evalPair` lbp)

16 >>=/ (\(a, b) -> wait (a + b + key))

17 where

18 x = sumPBT l (lbp -1)

19 y = sumPBT r (lbp -1)

図 28 sumPBT 関数

2. 5. 1 並列に全てのノードの和を計算
全てのノードの和を計算する場合，全てのノードを走

査する必要がある．このような処理は，並列化により高
速化が可能である．
並列に全てのノードの和を計算する sumPBT関数の定

義は，図 28のようになる．基本的には並列 fib関数や並
列 pow2関数と同様，再帰的に呼び出す部分を並列化し
て別のプロセッサに任せる．
sumPBT 関数の並列時間計算量は，pow2Par 関数の

並列時間計算量を表現する関数 pow2Costによって表現
できる．これは，1つ小さな値を引数として 2つ並列に呼
び出す pow2Par関数と，1つ高さが小さい完全二分木を
引数として 2 つ並列に呼び出す sumPBT 関数とが，同
じ構造をしているためである．pow2Cost 関数の引数と
しては，getHeight p-1を渡す．この定義では完全二分木
は高さ 1の時点で再帰を終えるので，完全二分木の高さ
から 1を引いた値を渡す．
2. 5. 2 並列に全てのノードの中から最大値を計算
並列に全てのノードの和を計算する maxPBT関数も，

sumPBT 関数と同様に，図 2. 5. 1 のように定義できる．
同じように全てのノードを走査する必要があるが，並列化
によって左右の部分木を別のプロセッサに担当させるこ
とができる．時間計算量も，sumPBT関数と同様である．

2. 6 リスト処理の並列分割
次に，ある程度長いリストに対して処理を行う場合に，

その処理を分割して並列化することで高速化を図るよう
な場合を考える．
countSolve関数は，文字列からなる数独の問題のリス

1 {-@ maxPBT :: { p:PBT Nat | getHeight p /= 0 }

2 -> lbp:Nat

3 -> { t:Tick Int |

4 (tcost t == pow2Cost (getHeight p - 1) lbp) }

5 @-}

6 maxPBT :: PBT Int -> Int -> Tick Int

図 29 maxPBT 関数

1 {-@ countSolve :: { xs:[String] | length xs /= 0 }

2 -> { t:Tick Int | tcost t == calcCost * length xs }

3 / [length xs]

4 @-}

5 countSolve :: [String] -> Tick Int

6

7 {-@ countSolveList :: { n:Nat | n > 0 }

8 -> { m:Nat | m > 0 }

9 -> { xs:[String] | length xs == n * m }

10 -> { t:Tick [Int] | tcost t == calcCost * n }

11 @-}

12 countSolveList :: Int -> Int

13 -> [String] -> Tick [Int]

14 countSolveList n m xs =

15 mapParList m ys (calcCost * n) countSolve

16 where

17 ys = chunk n m xs

18 c = calcCost * n

図 30 countSolve 関数，countSolveList 関数

トを受け取り，解を得られた個数を記録する関数である．
1つの数独の問題を解くのにかかるステップ数を calcCost

として，全体のステップ数は calcCostにリストの長さを
掛けたものとする．ここでは，countSolve関数の実装に
は踏み込まない．
countSolve関数をリストに適用すれば，逐次処理の結
果 Tick Int が得られる．これを並列化したのが count-

SolveList 関数である．並列化においては，mapParList

関数と chunk関数を用いることができる．まず，chunk

関数を用いて，与えられたリストを長さ nのm個のリス
トに分割する．この時，mとしてプロセッサ数（または
それ以下）の値を渡す．得た結果 ysは文字列のリストの
リストの型 [[String]]をもつので，mapParList関数に渡
して，並列に各要素 [String]を countSolve関数に渡して
評価させる．
分割したリストに対して並列に処理が行われるので，結
果のステップ数は 1つのプロセッサで行われる countSolve

関数のステップ数，すなわち calcCostに nを掛けた値と
なる．
このように，配列に対して処理を行い，Tickを返すよ
うな関数は，プロセッサ数などを渡して配列を分割する
ことで，比較的容易に並列化することが可能である．



1 par :: a -> b -> b

2 pseq :: a -> b -> b

図 31 par 関数，pseq 関数

3 並列時間計算量解析ライブラリの実装
本研究では，Liquid Haskellにおける計算量解析ライ

ブラリを，並列処理に拡張した実装を提案する．並列処
理関数としては，要素 2つのタプル（ペア），およびリス
トの形で渡された関数への処理を行う．

3. 1 Haskellにおける正格評価
Haskell において式を正確に評価する方法として，

deepseq 関数がある．deepseq 関数は，式を正規形ま
で評価する関数である．正規形とは，それ以上評価する
部分式のない値のことである．deepseq 関数は，引数の
型が NFData型クラスを満たしていることを制約とする．

3. 2 並列Haskell

Haskellにおける並列化の手法には，いくつか方法があ
る．その中でも，parallelパッケージを用いた手法につい
て説明する．
もっとも単純な並列性の実現として，Control.Parallel

の par関数と pseq関数を用いる方法がある．
par関数は，第 1引数と第 2引数を並列に評価するこ

とが有用であることを示し，第 2引数を返す関数である．
第 1引数に対応したスパークが生成され，スパークプー
ルに追加される．ここで，スパークとは，実行待ちの計
算処理のことであり，それらをためておく場所がスパー
クプールである．スパークプールに追加されたスパーク
の評価は，ワークスチール (work stealing)という手法を
用いて余ったプロセッサに割り当てられる．
pseq関数は，第 1引数を正格評価し，第 2引数を返す

関数である．ただし，正規形まで評価するわけではないの
で，正規形まで評価したい場合は deepseq関数を用いる．
上記の 2つの関数を用いても並列性は記述できる．し

かし，par/pseq関数は式の途中に書く必要があり，並列
性がアルゴリズムに依存してしまう．そこで，同じパッ
ケージに含まれる Control.Parallel.Strategies を紹介す
る．（図 32）
Strategy型は，どのような評価戦略を用いるかを表す型

である．using関数は，値と評価戦略を受け取り，その評
価戦略で値を評価する関数である．using型は中置演算子
として用いることができる．例えば(f x) `using` rpar

などと書くことで，値やそれを計算するアルゴリズムと，

1 infixl 0 `using`

2 using :: a -> Strategy a -> a

3

4 rdeepseq :: NFData a => Strategy a

5 rpar :: Strategy a

6

7 rparWith :: Strategy a -> Strategy a

8

9 evalTuple2 :: Strategy a -> Strategy b

10 -> Strategy (a, b)

11 parList :: Strategy a -> Strategy [a]

図 32 Control.Parallel.Strategies

評価戦略を分けて記述できる．
rdeepseq関数は正規形までの評価，rpar関数は並列に
評価を行う戦略を返す関数である．rparWith関数は，並
列評価に加えて評価を追加する．例えば，rparWith関数
に rdeepseq関数を渡すことで，並列に正規形まで評価で
きる．
evalTuple2 関数は，2 つの評価戦略を受け取り，1 要
素目と 2要素目にその戦略を適用するような評価戦略を
返す関数である．parList関数は，評価戦略を受け取り，
リストに対してその戦略を適用するような評価戦略を返
す関数である．
以降では，これらの関数を組み合わせて並列処理の実
装を行う．

3. 3 並列計算量解析関数の素朴な実装
Liquid Haskellにおける時間計算量解析を並列処理に
拡張することを考えてみる．まず具体的な状況として，並
列なフィボナッチ関数の時間計算量解析を試みる．
素朴な実装として，Tickモナドから tvalと tcostを直
接取り出して適切に操作を行う方法を選択する．並列な
フィボナッチ関数として実装した関数 fibParの定義は図
33のようになる．
得られる結果自体は 2. 3節で紹介した並列フィボナッ
チ関数と同一である．lbp = 0の場合は逐次処理，lbp > 0

の場合は並列処理を行うのも同一である．
一方で，2. 3節の場合と比べて tvalに関する推論が可
能になっている．tval t == fib nは，返された値が逐
次処理のフィボナッチ関数に同じ入力 nを与えた場合と
等しいことを篩型の制約として保証する．
しかし，fibPar関数の返す型を，Tick Natであるとす
ると推論に失敗する．フィボナッチ関数の値はNat，すな
わち非負の整数であることは推論できるべきであると予想
される．しかし，ここで問題となるのは，rdeepseq関数
の引数には，型クラス NFDataを満たしていることが要
求されることである．Nat型はあくまで Liquid Haskell



1 {-@ fibPar :: n:Nat -> lbp:Nat ->

2 { t:Tick Int |

3 (tcost t == fibCost n lbp) && (tval t == fib n) }

4 @-}

5 fibPar :: Int -> Int -> Tick Int

6 fibPar 0 _ = wait 1

7 fibPar 1 _ = wait 1

8 fibPar n 0 = Tick (c1 + c2 + c3) v3

9 where

10 x0 = fibPar (n -1) 0

11 y0 = fibPar (n -2) 0

12 Tick c1 v1 = x0 `usingStrategy` rdeepseq

13 Tick c2 v2 = y0 `usingStrategy` rdeepseq

14 Tick c3 v3 = wait (v1 + v2)

15 fibPar n lbp = Tick (c1 + c3) v3

16 where

17 x0 = fibPar (n -1) (lbp-1)

18 y0 = fibPar (n -2) (lbp-1)

19 Tick c1 v1 = x0 `usingStrategy` rparWith rdeepseq

20 Tick c2 v2 = y0 `usingStrategy` rdeepseq

21 Tick c3 v3 = wait (v1 + v2)

図 33 fibPar 関数

1 {-@ usingStrategy :: x:a

2 -> Strategy a -> {v:a | v == x } @-}

3 usingStrategy :: a -> Strategy a -> a

図 34 usingStrategy 関数

のアノテーション上の型であり，型クラス NFDataを満
たしているわけではないので，rdeepseq関数の返り値の
型は，引数が Nat ではなく Int として解釈されるため
Strategy Intが返り，usingStrategyを通して Intを返す．
usingStrategy関数は，第一引数を第二引数の評価戦略

に基づいて評価し，引数と同じ値を返す関数である．

3. 4 一般化の検討
先ほどの並列フィボナッチ関数の例では，Tickの中の

tcostを直接操作していた．しかし，tcostを直接操作す
ると，誤ってステップ数を増やしたり減らしたりしてし
まう場合が起こり得る．ここでは，ステップ数を適切に
処理してくれる並列処理関数を定義して，それを呼び出
すことを考える．
まずは，Tickを返す関数を引数に取った形での一般化

を試みる．具体的には，値を取って Tickを返す関数を 2

つと，それらへ渡す引数を受け取り，それらを実際に逐
次または並列に適用して，tcostを適切に加えた Tickを
返すような形である．しかし，Tickを返す関数に適用し
た場合の tcostを得るには更に高階な推論が必要となり，
bind関数の推論ができないのと同様の理由から上手く行
かなかった．
そこで，関数の代わりに Tick 自体を渡すことにして

1 {-@ evalPair :: t12:(Tick a, Tick b) -> lbp:Nat

2 -> { t:Tick (a, b) |

3 tcost t ==

4 if lbp == 0

5 then (tcost (fst t12) + tcost (snd t12))

6 else tcost (fst t12) &&

7 tval t == (tval (fst t12), tval (snd t12))}

8 @-}

9 evalPair :: (NFData a, NFData b)

10 => (Tick a, Tick b)

11 -> Int -> Tick (a, b)

12 t12 `evalPair` 0 = seqPair t12

13 t12 `evalPair` lbp = parPair t12 lbp

図 35 evalPair 関数

みる．具体的には，Tick から構成されるペアを受け取
り，ペアを要素として持つ Tick を返すような関数を考
えてみる．それが，本研究で実装した関数の 1つである，
evalPair関数である．

3. 5 並列にペアを処理する関数
3. 5. 1 evalPair関数
evalPair 関数の紹介は，2. 2. 1 節で行っている．ここ
では，内部の実装に関しての話を行う．evalPair関数は，
引数として取ったプロセッサの 2の対数 lbpが 0ならば
逐次処理関数 seqPairを，正ならば並列処理関数 parPair

を呼び出す．
並列 Haskell の性質から，並列化した場合も結果が変
わらないことは保証したいが，SMTソルバの推論の限界
から，evalPair関数では篩型の制約には含まれているも
のの実際に使う際には推論できていない．
3. 5. 2 seqPair関数
seqPair関数の定義は図 36 のようになる．seqPair関
数は，引数として Tick から構成されるペア xy0 を受け
取り，それを逐次的に評価した結果を返す関数である．
evalTuple2関数は，2要素のタプルの 1つ目と 2つ目の
評価戦略をそれぞれ受け取り，2 要素のタプルの評価戦
略を返す関数である．1つ目と 2つ目を rdeepseqで正規
形まで評価するとして，ペアの要素を逐次的に評価する．
評価した結果の t1と t2はそれぞれ Tick型なので，分解
してそれぞれの tcostを加算し，tvalはペアとして返す．
3. 5. 3 parPair関数
parPair関数の定義は図 37のようになる．parPair関
数は，引数として Tick から構成されるペア xyl とプロ
セッサ数の 2の対数 lbpを受け取り，それを並列的に評
価した結果を返す関数である．プロセッサ数の 2の対数
は，正であること，つまり空いたプロセッサが存在する
ことを確認するために用いており，処理には使用しない．
seqPair関数と共通する部分が多いが，1つ目の要素の評



1 {-@ seqPair :: t12:(Tick a, Tick b)

2 -> { t:Tick (a, b) |

3 tcost t == tcost (fst t12) + tcost (snd t12) &&

4 tval t == (tval (fst t12), tval (snd t12))}

5 @-}

6 seqPair :: (NFData a, NFData b)

7 => (Tick a, Tick b) -> Tick (a, b)

8 seqPair xy0 =

9 Tick (tcost t1 + tcost t2) (tval t1, tval t2)

10 where

11 (t1, t2) =

12 xy0 `usingStrategy`

13 evalTuple2 rdeepseq rdeepseq

図 36 seqPair 関数

1 {-@ parPair :: t12:(Tick a, Tick b)

2 -> { lbp:Nat | lbp > 0 }

3 -> { t:Tick (a, b) |

4 tcost t == tcost (fst t12) &&

5 tval t == (tval (fst t12), tval (snd t12))}

6 @-}

7 parPair :: (NFData a, NFData b)

8 => (Tick a, Tick b) -> Int -> Tick (a, b)

9 parPair xyl _ = Tick (tcost t1) (tval t1, tval t2)

10 where

11 (t1, t2) =

12 　　 xyl `usingStrategy`

13 　　 evalTuple2 (rparWith rdeepseq) rdeepseq

図 37 parPair 関数

価戦略として rdeeepseq の前に rparWith を加えて並列
処理を行っている点，tcostの計算に t1のみを使用して
いる点が異なる．
parPair 関数は，篩型の制約ではない暗黙の制約とし

て，ペアの 1つ目の要素の tcostが 2つ目の要素より大
きいことを仮定している．これは，結果の tcostとして，
1つ目の tcostを割り当てることで，推論を簡略化するた
めである．
より tcost が大きい方を採用する，といった実装にし

た場合について検討する．このような実装の場合でも，
pow2Par関数のように，左右の tcostが一致するような
場合は推論が行える．一方で fibPar関数は左右で tcost

が一致せず，さらに左右のどちらがより tcostが大きいか
を SMT ソルバでは推論できない．そのため，暫定的な
処置として，暗黙の制約を設けることを選択した．

3. 6 並列にリストを処理する関数
ペアに対して並列に処理を行う parPairと同様，リス

トに対して並列に処理を行う関数を考える．
3. 6. 1 parList関数
parList関数の紹介は，2. 2. 2節で行っている．ここで

は，内部の実装に関しての話を行う．

1 {-@ parList :: c:Nat

2 -> { xs:[{ t1:Tick a | tcost t1 <= c }] | xs /= [] }

3 -> { p:Nat | p >= length xs }

4 -> { t:Tick { os:[a] | length xs == length os } |

5 tcost t == c }

6 @-}

7 parList :: (NFData a)

8 => Int -> [Tick a] -> Int -> Tick [a]

9 parList c xs p =

10 Tick c (go xs `usingStrategy` parList rdeepseq)

11 where

12 {-@ go :: xs:[Tick a]

13 -> { ys:[a] | length xs == length ys } @-}

14 go :: [Tick a] -> [a]

15 go [] = []

16 go (x : xs) = cons (tval x) (go xs)

図 38 parList 関数

parList関数は，Tickのコストとして c，値としてはリ
スト xsを並列に実行した結果を指定する．，usingStrategy
関数と parList関数，rdeepseq関数を用いて，並列にリ
スト xsを正規形まで評価している．go関数は，Tickの
リストからリストを取り出す補助関数である．
コスト c の値は，Tick 型からなるリスト xs のうち，

tcostが最大のものを取ってくれば本来必要がないもので
ある．しかし，Liquid Haskellの推論において，そのよ
うな値を取り出して制約とすることは難しい．また，本
来なら xsを第一引数とした方が自然ではあるが，xsの全
要素が条件を満たしていることを明記するには，コスト c

を先に渡さざるを得ないため，このような順にしている．
例えば，受け取るリストを，特定の値以下の Tick型か
らなるリストを表すデータ構造とすれば，この問題は解
決できる可能性がある．しかし，単にコストと Tick型の
リストを受け取る現在の形の方が，他の処理と組み合わ
せる際には便利である．
3. 6. 2 chunk関数
chunk関数は，入力として正の数 nとm，長さ n×m

のリスト xsを受け取り，長さ nの m個のリストからな
るリストを返す関数である．篩型を用いて，指定した長
さとなるように指定している．
ここでは Tickに包まない単なるリストのリストとして
返している．しかし，並列に分割する部分も処理の一部
なので，tcostを記録して Tickに包んで返すべきである
可能性はある．
3. 6. 3 map関数
map関数は，コスト cと，そのコストとなる Tickを
返す関数 fと，リスト xsを受け取り，fを xsに適用した
結果として Tick型からなる配列を返す関数である．この
関数では，単にリスト xsの各要素に関数 fの処理を適用
するだけで，並列処理に基づいた挙動などは行わない点



1 {-@ chunk :: { n:Nat | n /= 0 }

2 -> { m:Nat | m /= 0 }

3 -> { xs:[a] | length xs == n * m }

4 -> { yss:[ { ys:[a] | length ys == n } ] |

5 length yss == m }

6 / [m]

7 @-}

8 chunk :: Int -> Int -> [a] -> [[a]]

9 chunk _ 1 xs = [xs]

10 chunk n m xs = cons as (chunk n (m -1) bs)

11 where

12 as = take n xs

13 bs = drop n xs

図 39 chunk 関数

1 {-@ map :: c:Nat

2 -> (a -> { t:Tick b | tcost t <= c })

3 -> xs:[a]

4 -> { ys:[{ t:Tick b | tcost t <= c }] |

5 length ys == length xs }

6 @-}

7 map :: Int -> (a -> Tick b) -> [a] -> [Tick b]

8 map c f [] = []

9 map c f (x : xs) = cons (f x) (map c f xs)

図 40 map 関数

に注意が必要である．コスト cを渡すことで，篩型の制
約として処理結果の Tick型の tcostが全て c以下である
ことを保証している．
3. 6. 4 mapParList関数
mapParList 関数は，parList 関数と map 関数を内部

で利用する関数である．正の整数 nと，長さ nのリスト
xs，コスト c，コスト cの Tick型を返す関数 fを受け取
り，xs の各要素に並列に関数 f を適用した結果として，
リストを tvalの型とするような Tickを返す関数である．
返り値の Tick型の tcostは，cと一致することを保証し
ている．
parList 関数の第二引数としては，map 関数を適用し

た後の Tick のリストを渡す．つまり，map 関数に渡し
た段階ではどのように評価されるかは決まらず，それを
parList関数の引数として渡すことで，並列に評価される
ことが決まる．map関数の返り値としての tcostは，捨
てられることとなる．

4 並列時間計算量解析の妥当性に関する議論
4. 1 実行時間測定結果との比較
本研究では並列プログラムのステップ数を解析する手

法を提案した．その妥当性の検証のため，実際に並列プ
ログラムの実行時間を計測して，解析によって得たステッ
プ数との比較を行う．本節での実行時間は，測定対象の

1 {-@ mapParList :: { n:Nat | n > 0 }

2 -> { xs:[a] | length xs == n }

3 -> c:Nat

4 -> (a -> { t:Tick b | tcost t <= c })

5 -> { t:Tick { os:[b] | length xs == length os } |

6 tcost t == c }

7 @-}

8 mapParList :: (NFData b)

9 => Int -> [a] -> Int -> (a -> Tick b) -> Tick [b]

10 mapParList n xs c f = parList c (map c f xs) n

図 41 mapParList 関数

計算が始まる時刻から，計算が終了するまでの経過時間
のことを指す．実行環境は，次の通りである．

CPU AMD Ryzen Threadripper 3970X

32-Core Processor

RAM 32.0GB

OS Ubuntu 20.04.2 LTS

GHCでは，同時に並列で走らせることができるスレッ
ド数を，実行時オプション-N で指定できる．この値は
CPUのコア数と等しくすることが推奨されている．今回
は 32 コアの CPU を用いるため，-N で指定する値，す
なわち想定するプロセッサ数を 1から 32まで変化させ，
それぞれ測定を行う．
4. 1. 1 完全二分木の走査
表 1は，完全二分木の走査における実行時間 3回分と
その平均，解析したステップ数の比較である．
ここでは，引数 n = 27，引数 lbpをプロセッサ数の 2

の対数として 2. 5. 1 節の maxPBT 関数の計算を測定対
象とした．ただし，完全二分木の生成は予め行い，実行
時間には含んでいない．
プロセッサ数としては 1,2,4,8,16,32を指定し，それぞ
れ引数 lbpとして 0,1,2,3,4,5を与えた．
また，横軸をプロセッサ数，縦軸を実行時間および解
析したステップ数としたグラフが図 42である．
グ ラ フ を 見 る と ，プ ロ セッサ 数 (numbers of

processors)を増やすごとに，解析したステップ数 (step)

とともに実行時間 (time) も減少傾向にあることがわか
る．一方で，プロセッサ数が 32個の場合の実行時間は，
プロセッサ数が 16個の場合より短くなっていない．
さらに，横軸を解析したステップ数，縦軸を実行時間
として，回帰直線を引いた散布図が図 45である．
実行時間とステップ数の相関係数は，0.9939287316だっ
た．ただし，データ数が 6つと非常に少ないこと，相関
があると判断できるからと言って有用性にすぐ結び付け
られるわけではないことに注意が必要である．



プロセッサ数 1回目の実行時間 (s) 2回目 (s) 3回目 (s) 平均実行時間 (s) ステップ数
1 = 20 10.987433 10.996685 11.023951 11.00269 134,217,727

2 = 21 7.094920 7.064267 7.036276 7.065154 67,108,864

4 = 22 4.063124 3.993696 4.005592 4.020804 33,554,433

8 = 23 2.457884 2.472770 2.487182 2.472612 16,777,218

16 = 24 1.869392 1.882695 1.882147 1.878078 8,388,611

32 = 25 2.216256 2.229249 2.196343 2.213949 4,194,308

表 1 完全二分木の走査における実行時間とステップ数

図 42 完全二分木の走査における実行時間とステップ数の比較

図 43 完全二分木の走査における実行時間とステップ数による
散布図

4. 1. 2 リスト処理の並列分割
表 2 は，リスト処理の並列分割における実行時間と，

解析したステップ数の比較である．
ここでは，16,000 個の数独の問題を，2. 6 節の map-

ParList 関数を用いてプロセッサ数で分割してそれらを
解く計算を測定対象とした．プロセッサごとに均等に割
り振れるよう，プロセッサ数として 1,2,4,5,8,10,16,20,32

をそれぞれ指定し，計測を行った．

図 44 リスト処理の並列分割における実行時間とステップ数の
比較

図 45 リスト処理の並列分割における実行時間とステップ数に
よる散布図

また，横軸をプロセッサ数，縦軸を実行時間および解
析したステップ数としたグラフが図 44である．
さらに，横軸を解析したステップ数，縦軸を実行時間
として，回帰直線を引いた散布図が図 45である．
実行時間とステップ数の相関係数は，0.9998209612だっ
た．こちらも同様にデータ数が少なく，有用性に直接結
びつけるには注意が必要である．



プロセッサ数 1回目の実行時間 (s) 2回目 (s) 3回目 (s) 平均実行時間 (s) ステップ数
1 13.727178 14.910259 13.647531 14.09499 1.6× 109

2 7.563414 7.460836 7.429231 7.484494 8.0× 108

4 4.087452 4.191272 4.068193 4.115639 4.0× 108

5 3.335491 3.307790 3.386718 3.343333 3.2× 108

8 2.330277 2.402992 2.358420 2.363896 2.0× 108

10 1.926612 1.933136 1.911500 1.923749 1.6× 108

16 1.356350 1.326441 1.350727 1.344506 1.0× 108

20 1.289420 1.195688 1.184411 1.223173 8.0× 107

32 1.223408 0.965334 1.026178 1.071640 5.0× 107

表 2 リスト処理の並列分割における実行時間とステップ数

4. 2 PRAM model採用の妥当性
本研究では，並列計算のモデルとして PRAM を採用

した．PRAMは，並列アルゴリズムの時間計算量や空間
計算量を考えるための抽象的な計算モデルで，複数台の
RAMを並列に動かすモデルである．RAMは，メモリの
どこにでも一定時間でアクセスが可能とする計算モデル
である．しかし，Haskellのメモリモデルでは，ランダム
アクセスが一定時間で行えるわけではない．
並列計算のモデルとしては，有向非巡回グラフモデル

（以下 DAG モデルと呼ぶ）がある．DAG モデルでは，
並列 Haskell と同様，タスクをプロセッサに割り振るス
ケジューリングの問題を抽象化している．一方で，それ
によりアルゴリズムの時間計算量の解析が困難である．
本研究では，時間計算量の解析を容易にするために

PRAMを採用したが，他の計算モデルの採用は検討の余
地がある．

5 おわりに
5. 1 結論
本研究では，Liquid Haskellにおいて，並列プログラ

ムの時間計算量を静的に解析する手法を提案し，これを
実装した．また，提案した実装を適用して，いくつかの並
列プログラムにおける時間計算量の解析を行った．実際
に並列プログラムを実行して経過時間の計測を行い，解
析したステップ数と比較してある程度妥当であることを
確認した．

5. 2 今後の課題
今後の課題としては，本研究で提案した手法をより多

くの並列プログラム例に対して適用することが挙げられ
る．クイックソートやマージソートのような典型的なア
ルゴリズムから，より実用的なプログラムまで様々な適
用先を考えることができる．
また，Liquid Haskellでは，Proof Combinatorsを用

いて，自動推論の制約に囚われず手動で証明を行うこと
もできる．Proof Combinatorsを利用した場合の並列プ
ログラムの時間計算量の解析についても，今後の課題と
する．
さらに，議論の章で述べた通り，PRAM以外の計算モ
デルの採用も検討の余地があると思われる．
謝辞 本研究においてご意見いただいた西崎研究室の皆

